
1 Introduction

This article reports my experience on using Julia for competing in Al Zimmermann's �Delacorte Numbers�
Programming Contest. The resulting code won 2nd place using only the computing power of a quad-core
processor. The key features that made Julia shine for this contest were:

Interactivity Julia's interactive �read-eval-print loop� (REPL) let me investigate data quickly on whim.
As I went along, I grew a set of utilities for investigation.

Quick Prototyping Julia's high-level features and rich standard library let me quickly prototype numerous
algorithms, without getting caught up in low-level details such as memory management. Most ideas I
came up with failed, so fast prototyping to try them and move on was critical.

Fast Code Julia's support for e�cient code let me implement scalar and SIMD code for the ideas that
proved useful, without having to recode in another language, or losing the interactive environment.

This article describes the �nal program in detail and gives tips on writing high-performance Julia. For
newcomers, the discussion contains tutorial information in boxes labeled �Julia mechanics�. Experienced
Julia programmers can skip the boxes.

Disclaimer: I work on the Julia compiler and contributed its @simd feature. My perspective may be
biased.

2 The Problem

See the contest website1 for a comprehensive description of the contest problem. The problem was to arrange
the integers 1...N2 in an N×N matrix in a way that either minimizes or maximizes the value of an objective
function. The value of N ranged from 3 to 27, so there were a total of 50 problems, 25 maximization problems
and 25 minimization problems. To avoid confusion with other matrices in the code, the rest of this paper
calls the arrangement a �Delacorte matrix�.

The objective function is the �Delacorte Number�, an invention of Al Zimmermann. It is de�ned as:

Delacorte Number =
∑
u,v

gcd(u, v) · distance2(u, v) (1)

where the sum is taken over all possible unordered pairs u, v of matrix elements where:

• gcd(u, v) denotes the greatest common divisor of the values of u and v.

• distance(u, v) denotes the Euclidean distance between the coordinates of u and v, assuming the matrix
is laid out on a unit grid.

1 Which is http://trdb.org as of 2015-1-20, but will later move to http://azspcs.com.

The minimization problem corresponds closely to a classic problem of placing components on a circuit board
in a way that minimizes wire cost, where the number of wires between two components u and v is gcd(u,v).
The di�erence from the classic problem is that the Delacorte number uses the square of the distance, not
the distance. Interestingly, some heuristics for the classic placement problem use the square of the distance
because it enables much simpler math than using the distance. Look up �Quadratic Placement� for details.
The Delacorte number problem is a special case of the Quadratic Assignment Problem.

It's not clear whether the maximization problem has a practical use, though during the contest, I noted
that it enabled an urban planner to arrange urban sprawl in a way that maximizes tra�c congestion in a
city. I lived there once.

Here is Julia code for computing the Delacorte number of a Matrix a:

# Slow Delacorte calculation used for checking
function slowDelacorte{T<:Integer}( a::Matrix{T} )

(m,n) = size(a)
s = 0
for x0=1:n, x1=1:n, y0=1:m, y1=1:m

s += gcd(a[y0,x0],a[y1,x1]) * ((x0-x1)^2 + (y0-y1)^2)
end
@assert iseven(s)
div(s,2)

end

Julia Mechanics: Julia keywords are shown in blue. Julia standard library identi�ers are shown in brown.
Function slowDelacorte takes any matrix a with elements of type T, which can be any subtype of abstract
type Integer. Variables m and n are the dimensions of matrix a. The for loop is essentially four nested
loops, with index variables x0, x1, y0, and y1. A function implicitly returns the last expression that it
evaluates. This function returns bs/2c.

To simplify bookkeeping, slowDelacorte sums pairs redundantly. For example it tallies both (u, v) and
(v, u), and includes self-pairs (u, u). The self-pairs do not a�ect the sum since their distance is zero. The
division by two at the end compensates for tallying each pair twice. The code emphasizes simplicity over
e�ciency because it is used only for checking faster algorithms, and so correctness is paramount.

3 Sketch of Algorithm

I evolved the program over time by trying various things manually, and keeping the ones that worked.
Particularly in early development, there was a tinge of gambling excitement in choosing parameters, �rolling
the dice�, and seeing what happened. Eventually I would tire of the manual typing and would automate a
step. One level of the algorithm is even called robo because it robotically executes a recipe that had been
developed by much manual experimentation. Then the experimentation was repeated at a higher level. In
the �nal program, call chains recapitulate phylogeny.

The program started out as a greedy algorithm that created a random solution and tried swapping
random pairs of elements, accepting swaps that improved it. Alas for this problem, the greedy algorithm
easily gets stuck in a local optimum. Repeating this algorithm many times in hope of getting lucky helped
a little, but not enough to be competitive.

What works better is a form of simulated annealing, which is like the greedy algorithm, but also accepts
swaps that make the solution worse. That lets it hop away from a local optima that might otherwise trap it.
Of course, it might hop away from a global optimum too! So it accepts worse solutions only if the worsening
is less than a �temperature� threshold. Annealing starts with a high temperature and slowly cools it over
time. The high temperatures coarsely sample large portions of the search space. The low temperatures �ne
tune the solution within a small portion of the search space.

The simulated annealing is augmented by running multiple copies of annealing in rounds, and letting the
best solutions in each round advance to the next round.

Quickly evaluating swaps is critical to performance. It takes timeO(N4) to evaluate the sum in Equation 1
directly, and O(N2) time to evaluate only the contribution to it from two swapped elements. The program
does much better, taking only O(1) time to evaluate how much a swap a�ects changes a Delacorte number.
Accepting a swap takes almost O(N2) time, which is not a serious bottleneck, because most swaps are
rejected and the O(N2) work vectorizes. 2 Section 4.6.1 supports this claim with pro�le data.

At the next level up, the algorithm evolves a population of candidate solutions. The initial population
is seeded with roboPop copies of a seed matrix, Each sample in the population is subjected to roboArity
separate runs of swapping while exponentially cooling the temperature. Then the cooled copies are sorted,
and a new population created from the best roboPop results.

The entire process above was iterated, starting at a high initial temperature, and then adjusting that
initial temperature for each cooling as the population evolved. Furthermore, each of the roboArity copies
was started at a di�erent temperature, spread over a window with a 0.7 ratio3 from high to low ends, for
reasons explained in Section 4.7. After each processing of the entire population, the window was adjusted
up or down. The rule of thumb was to adjust it so that the highest temperature that had actually improved
an individual became the center (logarithmically speaking) of the temperature window for the next round.

The �rst round was treated di�erently than the rest of the rounds. In the �rst round, the temperature
was typically much higher, in order to �kick� the samples way from the known solution. This round never
included the original samples in the resulting population. The rest of the rounds added the old samples to
the population before sorting and culling, so that the best solutions found so far would not be lost.

4 Julia Program

The code excerpts assume that N is already de�ned. Typically the program was run from a fresh interactive
session by initializing N, including the program text, and running a routine therein. For example:

julia> const N=17; include("delacorte.jl")
N=17
weight built
indexPairArray built

julia> flog(up,100)

Julia Mechanics: The const says that N cannot be rebound to another object, which helps the compiler
optimize. It turns out to be an insigni�cant performance gain here because N is not referenced directly by
compute-intensive portions of the program. However, the const is nonetheless a good idea here because the
rest of the program assumes that N never changes. The end of Section 4.3 says more about const.

When the program �nds a publishable improvement, it saves the solution, printed an announcement in the
log, using a distinct color so it was easy to �nd later, and emits a bell ding (\a) to note the happy news.
Those encouraging dings were down to a few a day near the end of the contest.

4.1 Julia Version Compatibility

The examples assume Julia 0.4, which is under development and subject to change. Most of the incom-
patibilities with the production release Julia 0.3.5 can be worked around by �rst loading Julia's version
interoperability package Compat.jl, like this:

julia> Pkg.add("Compat")
INFO: Initializing package repository C:\Users\Dad\.julia\v0.3
INFO: Cloning METADATA from git://github.com/JuliaLang/METADATA.jl
INFO: Cloning cache of Compat from git://github.com/JuliaLang/Compat.jl.git
INFO: Installing Compat v0.2.9

2 Throughout this article, �vectorizes� means the code uses SIMD instructions e�ciently. Regrettably, �vectorized code� has
another meaning to Julia programmers: code written using vector operations, like APL or Fortran90 array-expression style.

3 An approximation of
√
2. Given the level of empirically derived hackery, it seemed silly to use more precision.

INFO: Package database updated

julia> require("Compat")

julia> include("delacorte.jl")

Once you have added the package, you can skip the add step in future sessions.
However, at the time of this writing, Compat.jl does not support the notations used to de�ne function

fast_int32 in Section 4.2. That section prescribes a suitable de�nition.

4.2 Working Arithmetic Type

The fast method for evaluating swaps depends on some lookup tables and vectorized loops. It was not clear
to me what numeric type would yield the best vectorization since compilers and instruction sets can be
quirky about the matter, so the type is declared as a global constant QType. Some of the code needs to use
Int32 regardless, because a Delacorte number won't always �t precisely in a Float32. 64-bit types were
not considered, since they would double the memory tra�c and halve the number of scalars processed by a
SIMD instruction.

Here is the de�nition of QType and fast conversions to Int32:

1 # Numeric type used for Delacorte calculations. It should be a signed type big
2 # enough to hold a coordinate or 1st moment. Int32 or Float32 are good choices.
3 const QType = Int32
4

5 # De�ne fast conversions to Int32.
6 fast_int32(x::Integer) = x % Int32
7 fast_int32(x::FloatingPoint) = Base.unsafe_trunc(Int32,x)

Julia Mechanics: Lines 6 and 7 show the short form syntax for de�ning functions in Julia. Function
fast_int32 has two de�nitions, one for integer arguments and one for �oating-point arguments. The
expression x % Int32 returns a value congruent to x modulo 232. The function unsafe_trunc converts a
�oating-point value to an integral type without checking whether it �ts. Both of these typically compile to
a no-op or single instruction, depending on the actual argument type.

An alternative would be to use the Julia function int32 instead of fast_int32. But that would introduce
overhead because int32 checks that the conversion preserves the algebraic value, and if not, throws an
exception InexactError.

The contest version actually used trunc(Int,x) and int32, instead of fast_int32(x). At the time I was
under the mistaken impression that trunc(Int,x) elided checking.

If using Julia 0.3, replace lines 6 and 7 with this:

fast_int32(x) = itrunc(x) # Julia 0.3

Measurements on a single thread indicate that QType=Int32 makes the code run slightly faster. On some
architectures, the QType=Float32 could theorectically help by enabling instruction-level parallelism over
both integer and �oating-point functional units. But on my CPU, the integer units are a superset of the
�oating-point units, so QType=Float32 cannot add instruction-level parallelism. However, it does consume
more power � the machine runs measurably cooler when using QType=Int32.

4.3 Lookup Tables

The program spends most of its time evaluating the change in a Delacorte number if two elements of the
Delacorte matrix are swapped. Most swaps are rejected because the change is too severe. Using algebra, I

derive a method that takes O(1) time to evaluate a swap and almost O(N2) time to accept it.4 Since most
swaps are rejected, this is a big improvement over using the direct O(N4) formula for computing a Delacorte
number. Furthermore the O(N2) work vectorizes well.

The fast method for evaluating swaps relies on four lookup tables, computed at program startup, and
invariant afterwards. The tables are de�ned as follows:

• encode[u] maps a Delacorte matrix value u to its index in the other tables.

• decode[i] maps an index i to the corresponding matrix value.

• weight[i,j] =

{
gcd(decode[i], decode[j])− 1 if i 6= j

0 otherwise

• weightSum[j] =
∑
i

weight[i, j]

Making the weight �gcd-1� increases the number of zeros, which reduces the number of arithmetic operations
later.

The index encoding puts weightSum in descending order. That order has both a computational and an
interactive bene�t. The computational bene�t was that it allowed moving the all-zero rows of weight to
the end, where they could be represented implicitly, and easily omitted from calculations where they were
irrelevant.5 The interactive bene�t was that it put the values with strong connections where they were easy
to inspect. For example, evaluating decode[1:8] shows the eight most strongly connected elements, which
were always (from what I recall) the elements to use in the corners when shooting for a maximal Delacorte
number.

Here is the Julia code for initializing the tables:

1 # Compute weight, weightSum, and decode tables.
2 function makeWeight()
3 n = N^2
4 t = QType[i6=j ? gcd(i,j)-1 : 0 for i=1:n, j=1:n]
5 s = QType[sum(t[:,j]) for j=1:n]
6 e = sortperm(s,alg=MergeSort,rev=true)
7 permute!(s,e)
8 m = findlast(s)
9 w = QType[t[e[i],e[j]] for i=1:m, j=1:n]

10 (w,s,e)
11 end
12

13 # These three tables are indexed by encoded values.
14 const (weight,weightSum,decode) = makeWeight()
15

16 # Table of encodings
17 const encode = invperm(decode)

4 After the contest ended, I learned that "Whitefox" had published a similar method on the Russian web site �dxdy� during
the contest To �nd it, look at http://dxdy.ru/topic88308.html around November 21, 2014.

5 The table is symmetric, so the all-zero columns could be omitted too. But that turned out to complicate logic elsewhere
without any performance bene�t. Likewise, fully exploiting symmetry would let the table occupy half as much memory, but
would likely hurt performance because table accesses in critical loops would no longer have unit stride.

5

http://dxdy.ru/topic88308.html


Julia Mechanics: Line 4 constructs a 2D array of type QType. The array is the weights indexed by
unencoded values. Line 4 also demonstrates using a Unicode math symbol 6= instead of !=, which Julia also
accepts. Line 5 computes column sums. Line 6 sorts those column sums in descending order. Line 7 puts
the sums in sorted order. Line 8 �nds the last non-zero element of the column sums. Line 9 constructs the
weight table using the encoded ordering. Line 10 returns a tuple of the three tables of interest, and line 14
assigns the three tables to corresponding global constants.

Line 17 exploits a coincidence that both Julia array indices and the numbers in a Delacorte matrix start at
1. Hence the decode mapping looks like a permutation, and its inverse can be computed using a standard
library function for inverting permutations.

The code demonstrates concise code where the Julia standard library does most of the work. The only extra
function I had to write was findlast, since the Julia 0.4 standard library did not have it until the last two
days of the contest.

The const annotations on the tables are important for performance. Later parts of the code access them
heavily, and the annotations let the compiler know those variables never change type, and thus the compiler
can eliminate run-time type checks.6

4.4 Data Structure

The key data structure is a Quad. It inverts the representation of a Delacorte matrix: instead of holding
the elements, it holds their coordinates. Indeed, the elements are considered to be points in 2-space. Like
the earlier tables, it's indexed by encoded values. Additionally, a Quad holds the Delacorte number of the
matrix, and an array of ��rst moments�. These moments are what enable O(1) calculation of the change in
a Delacorte number caused by swapping two elements.

Here is the type declaration:7

1 abstract Delacorte
2

3 type Quad <: Delacorte
4 z :: Matrix{QType} # Coordinates of points
5 c :: Matrix{QType} # First moments of points
6 d :: Int # Delacorte number
7 # Construct from coordinates of points
8 function Quad( z::Matrix{QType} )
9 c = computeMoments(z)

10 new(z, c, computeDelacorte(z,c))
11 end
12 end

Julia Mechanics: Line 1 declares an abstract type. Line 3 declares a concrete subtype Quad, with three
�elds declared on lines 4-6. The �inner constructor� on lines 8-11 computes the moments from a matrix
of coordinates (line 9), and then returns a newly constructed Quad object (line 10). The correspondence
between the arguments to new and �elds of Quad is positional. Note that the c on line 9 is a local variable,
not the �eld declared on line 5. The call to new binds the �eld c to the same object as the local variable c.

Higher level parts of the program operate on the abstract type Delacorte. That way, I could easily plug
in a better representation later.8 Arrays z and c are n × 2 arrays of QType. The two columns correspond

6 Note for C/C++ programmers: const in Julia means something di�erent than const in Julia. In C/C++, a variable
refers to a memory location, and a const-quali�ed type prohibits modifying the memory location. In Julia, variables refer to
objects, and const prohibits rebinding the name to a di�erent object. The underlying object can still change unless it has
what Julia calls an immutable type, such as Int. There is work in progress on adding immutable arrays to Julia.

7 In this and some other listings, some of the de�nitions shown close together are actually far apart in the full program
delacorte.jl because of other intervening de�nitions.

8 A corollary of Murphy's law is that if you design software for future modi�cation, it won't change in that direction, but
instead need to change in a di�erent direction for which you didn't plan. I didn't think of a better representation. But Tomas
Rokicki did, in a way that enabled O(log(n)) evaluation or commitment of a swap.

respectively to the �x� and �y� positions in the conceptual matrix. Component d holds the Delacorte number
of the conceptual matrix.

4.4.1 First Moments

Now the ��rst moments� will be explained. they are a matrix product weight·z. Here is the code:
1 # Compute Delacorte "�rst moments" of points in z
2 function computeMoments{T<:Real}( z::Matrix{T} )
3 w = weight
4 (m,n) = size(w)
5 @assert m≤n
6 @assert size(z)==(n,2)
7 c = zeros(T,n,2)
8 @inbounds for i=1:m
9 Σx = zero(T)

10 Σy = zero(T)
11 @simd for j=1:m
12 Σx += w[j,i]*z[j,1]
13 Σy += w[j,i]*z[j,2]
14 end
15 c[i,1] = Σx
16 c[i,2] = Σy
17 end
18 c
19 end

Julia Mechanics: zeros(T,n,2) builds an n× 2 array of T �lled with zeros. Σx and Σy demonstrate the
use of Unicode in identi�ers. @inbounds and @simd are explained in the main text.

The subscripts run over 1:m instead of 1:n since weight(j, i) is zero if j or i exceeds m. As noted earlier,
only the �rst m rows of the table actually exist. The subscripts for w appear to be in the wrong order a
matrix multiplication, but that's okay since w[1:m,1:m] is symmetric.

The code computeMoments shows four tricks to writing fast Julia loops:

Type-stable code The initializations on lines 9 and 10 use zero(T), not 0. The reason is that 0 has type
Int, but T might be some other type. If the types are di�erent, then Σx would have type Int until
updated with +=, afterwards it might have a di�erent type. The changing type requires much more
complicated code generation. Hence it pays to ensure that accumulation variables start out with the
right type, and keep that type. Some ways to check are described later in this section.

Inner loop runs down columns The loops and subscripts are arranged so that the innermost loop runs
down columns of matrix w. That's more e�cient than running across rows because Julia uses Fortran-
style column-major storage, and it's much faster to access consecutive adjacent memory locations than
ones far apart.

@inbounds This macro lets the compiler presume that all subscripts are in bounds. That not only elim-
inates the direct instruction cost of checking subscripts, but also enables the compiler to be more
aggressive about reordering instructions.

Care is required. Without the checking, an out of bounds access can corrupt the entire system, so I
was careful to add enough assertions (lines 5 and 6) to protect against crashing an interactive session.
Maybe in the future the Julia compiler will be smart enough to infer from those assertions that the
bounds checks are super�uous.

@simd This macro grants the compiler permission to reorder evaluation of the loop in a way that enables
vectorization. Vectorization for the target platform requires that the array accesses have unit stride.
That's another reason for swapping the subscripts on w, and the reason for organizing z as two columns
instead of two rows.

In this example, it turns out that @simd is super�uous if the arithmetic is on Int32, because the inbounds
gives the compiler most of what it needed to know, and it inserts a cheap run-time check to extract the rest
of what it needs to know. But when the arithmetic is on Float32, @simd greatly helps because it grants
permission to reorder the reductions into Σx and Σy in a way that enables vectorization. Why the di�erence?
Integer addition is commutative and associative, so the Julia compiler is free to reorder such addition. IEEE
�oating-point addition is commutative but not associative. Julia is picky about precise �oating-point unless
the programmer writes otherwise.

The net speed improvement for computeMoments when N=27 is:

• 3.7x by using @inbounds when using Int32 arithmetic.

• 3.5x by using @inbounds and @simd when using Float32 arithmetic.

The annotation @simd does not mean �make it go faster�. It is a contract where you promise things to the
compiler. Do not sign the contract unless you understand it. See the article �Vectorization in Julia� to learn
more about what kind of deal you make with @simd.

The code does not declare the types of local variables. Doing so usually just adds clutter, and possibly
overhead of additional run-time type checking. To get speed, the trick is to write the code so that Julia's
type-inference engine deduces unique concrete types for each local variable. The Julia function code_typed
shows you the deduced types. A Union type therein indicates an ambiguity that can hurt performance. I'm
fond of using the Julia functions code_llvm and code_native to look at the code generated for critical
loops. See Leah Hanson's article �Julia Introspects� on how to use these functions. The package TypeCheck.jl
can �nd type instabilities in non-parametric functions.

In retrospect, not much time is spent in computeMoments, so the e�ciency tricks were of little value
here. Lines 8-17 could have been replaced by a single line that does a matrix multiplication:

c[1:m,:] = w[1:m,1:m]*z[1:m,:]

Julia Mechanics: c[1:m,:] denotes a sub-array containing the �rst m rows of matrix c.

Nonetheless I explain the tricks here because they are part of the code, and they are used later in time-critical
code.

4.4.2 Computing Delacorte Number from First Moments

Given the moments, the Delacorte number can be computed as follows:

1 # Compute sum over all distances squared
2 function makeSumOfPairDistances()
3 s = 0
4 for x0=1:N, x1=1:N
5 s += (x1-x0)^2
6 end
7 s*N^2
8 end
9

10 # Sum of all pair distances
11 const sumOfPairDistances = makeSumOfPairDistances()
12

13 # Compute Delacorte number from coordinates z and moments c.
14 function computeDelacorte{T<:Real,U<:Real}( z::Matrix{T}, c::Matrix{U} )
15 (m,n) = size(weight)
16 s = 0
17 @inbounds @simd for i=1:m
18 s += fast_int32(weightSum[i]*(z[i,1]^2+z[i,2]^2) - (c[i,1]*z[i,1] + c[i,2]*z[i,2]))
19 end
20 s + sumOfPairDistances
21 end

The addition of sumOfPairDistances on line 20 compensates for the -1 bias in the de�nition of table weight.
The rest of the proof that it computes a Delacorte number is an exercise for the reader.9 A Float32 does
not have enough precision to hold a Delacorte number accurately, hence the accumulator s is an Int. The
fast_int32 operation on line 18 deals with cases where �oating-point is used.

4.4.3 Abstract Access to a Quad

The program de�nes some auxiliary functions that enable Quad to act much like other Julia types:

import Base.copy, Base.getindex, Base.isless, Base.length, Base.size

# Get Delacorte Number of a Quad
delacorte( q::Quad ) = q.d

# Number of points in the Quad
length( q::Quad ) = size(q.z,1)

# Get coordinate of point at index i
getindex( q::Quad, i::Integer ) = (q.z[i,1],q.z[i,2])

# Get copy of q
copy( q::Quad ) = deepcopy(q)

# Size of rect(q)
size( q::Quad ) = (int(maximum(q.z[:,2])),int(maximum(q.z[:,1])))

# Order for sorting Quads by Delacorte value
isless( q::Quad, r::Quad ) = q.d<r.d

Julia Mechanics: The import is required so that the compiler considers the function as overloads of
functions in the Julia standard library instead of as functions that coincidentally have the same name. The
compiler is usually good about warning you if you forget the import.

A call getindex(q,i) can be invoked with the syntactic sugar q[i], which returns an (x, y) pair of coor-
dinates for the ith point. That and the overload of length make Quad quack like an array of points.

4.4.4 Printing and I/O

The Quad representation is awkward for human reading. So the program de�nes how Julia should print a
Quad. Here is sample showing how it looks in an interactive session:

julia> q = randQuad()
Size = 3 x 3
Delacorte# = 142
(1, 6, 5),
(7, 4, 3),
(8, 2, 9)

The format of the matrix part was suitable for copy-pasting into the contest submission site. The formatting
is achieved by overriding the standard library function show, like this:

9 The key lemma in the proof is that given a symmtric m×m matrix w, the following equality holds:

1

2

∑
i

∑
j

(xi − xj)
2wi,j =

∑
i

x2
i

∑
j

wi,j

−∑
i

xi

∑
j

wi,jxj


The left side is essentially the x-axis contribution in the slow Delacorte calculation. The rightmost sum is the x-axis portion of
the ��rst moment� calculation in computeMoments.

1 import Base.show
2

3 # Show Delacorte matrix
4 function show( f::IO, r::Delacorte )
5 println(f,"Size = $N x $N")
6 d = delacorte(r)
7 a = rect(r)
8 e = slowDelacorte(a)
9 if e==d

10 println(f,"Delacorte# = ",d)
11 else
12 println("ERROR: slow Delacorte# = ",e," fast Delacorte# = ",d)
13 end
14 printAZPC(f,a)
15 end

Julia Mechanics: Line 5 uses string interpolation to print N. It's not the most e�cient way to print a
value since it involves string construction, but it's convenient when performance is unimportant.

Function show is de�ned on the abstract type Delacorte instead of the concrete type Quad. The functions
delacorte and rect (de�nitions not shown) link the abstract to the concrete.

• delacorte extracts the delacorte number.

• rect constructs the intuitive matrix representation.

Since the Delacorte number extracted from r was computed by accumulating many deltas, and just one
wrong delta would corrupt the answer, the code checks it against a slow and simple calculation. Function
printAZPC takes care of formatting the matrix. See the full program for more details on these functions.

Results were stored in this format. It's not the simplest format to parse, but Julia's support for regular
expression matching made it straightforward. This is not earth-shaking since many modern programming
languages have support for regular expressions. But how many language also vectorize loops and can be
used interactively? See function readArray in the full program for details.

4.5 Evaluating and Committing Swaps

This section describes theO(1) algorithm for evaluating swaps. Look back at lines 17-19 of computeDelacorte.
Suppose the Delacorte matrix element with coordinate z[i,:] changes position. The impact on the sum s
can be computed by evaluating how the right side of line 18 changes for iteration i.

1. The values of z[i,1] and z[i,2] change.

2. The values of c[i,1] and c[i,2] change. However, these are dot products of a column of weight
and column of z, so their change can be found in O(1) time from the change in z[i,:].

The rest is basic algebra of �nite di�erences. A swap involves two complementary changes, and the comple-
mentary nature makes even more stu� algebraically cancel.10

1 # Compute change in delacorte(q) if numbers with encodings i and j are swapped
2 function delta( q::Quad, i::Integer, j::Integer )
3 (m,n) = size(weight)
4 z = q.z
5 c = q.c

10 Like many mathematicians, I'm presenting the simplest derivation, not how I originally derived it. Originally, I �rst
�gured out the calculations by algebraically analyzing the e�ect of a swap on a Delacorte number, and spotting how a common
subexpression (which became c) remained largely invariant. Implementing computeDelacorte in terms of c actually came
later.

6 @inbounds begin
7 xi = z[i,1]; yi = z[i,2]
8 xj = z[j,1]; yj = z[j,2]
9 ∆x = xj-xi

10 ∆y = yj-yi
11 p = (weightSum[i]-weightSum[j])*(∆x*(xj+xi)+∆y*(yj+yi)) -
12 2((c[i,1]-c[j,1])*∆x + (c[i,2]-c[j,2])*∆y)
13 if i≤m && j≤m
14 p += 2weight[i,j]*(∆x^2 + ∆y^2)
15 end
16 end
17 fast_int32(p)
18 end

Julia Mechanics: The factors 2 on lines 12 and 14 shows a small syntactic convenience: the multiplication
symbol can be elided when its left operand is a literal.

Committing a swap requires updating all of the elements of c that are not always zero, which takes O(m)
time, where m is slightly less than N2. Here is where judicious use of @inbounds and @simd pay o� for
vectorizing the loop.

# Swap encodings i and j. Caller must provide delta(q,i,j) as the last parameter.
function swap!( q::Quad, i::Integer, j::Integer, delta::Integer )

w = weight
(m,n) = size(w)
q.d += delta
c = q.c
z = q.z
xi = z[i,1]; yi = z[i,2]
xj = z[j,1]; yj = z[j,2]
z[i,1] = xj; z[i,2] = yj
z[j,1] = xi; z[j,2] = yi
# Update moments
∆x = xj-xi
∆y = yj-yi
@inbounds @simd for k=1:m

∆w = w[k,i]-w[k,j]
c[k,1] += ∆w*∆x
c[k,2] += ∆w*∆y

end
end

The code could defer the update of c until several swaps need to be committed, by keeping track of which
values of c had pending updates and only resolving updates for elements of c used by function delta. But
doing so seems counter-productive. Invocations of delta dominate invocations of swap!, so it pays to keep
delta as lean as possible.

4.6 Hopping and Cooling

The next level up in the code implements simulated annealing with a threshold acceptance test. Not shown
is a function getRandomPair that gets a pair of indices to swap, by looking up a random position in a table
of index pairs. Ideally, it would return the pair of indices as a tuple (i, j). Unfortunately, Julia currently
stores arrays of tuples ine�ciently, so I created my own type IndexPair. The table omits indices of values
that are pinned down (Section 4.9), and omits pairs of indices that are practically indistinguishable.11

11 Speci�cally i and j such that both weightSum[i] and weightSum[j] are zero, and hence a swap will not a�ect the
Delacorte number.

Function hop! does random swapping with an acceptance test. The name comes from a metaphor of a
frog hopping through solution space, looking for a minimum or maximum.

# Optimization directions
const up = Forward
const down = Reverse

# Try numIter random swaps. Accept those with delta no worse than temp
function hop!( r::Delacorte, o::Ordering, temp::Integer, numIter::Integer )

@assert o==up || o==down
threshold = o==up ? -temp-1 : temp+1
for trial=1:numIter

p = getRandomPair()
d = delta(r,p.i,p.j)
if lt(o,threshold,d)

swap!(r,p.i,p.j,d)
end

end
end

Julia Mechanics: An informal Julia convention is to add the su�x ! to functions with side e�ects.
Standard library function lt is a �less than� predicate whose �rst argument indicates the ordering direction.
Here the ordering is parameter o, which has di�erent types for the max vs. min problems. The Julia compiler
generates customized versions of hop! for each instance of parameter types, like the way a C++ compiler
instantiates a template for di�erent types of arguments.

When temp==0, the code nonetheless accepts swaps with zero delta. I don't know if that helped or not,
but the intuition was that there was nothing to lose by zero-gain changes might help poke around the search
space more.

Early during development I had rejected exponential cooling because it did not seem to work. In ret-
rospect, the early experiments were probably done with a cooling curve made too fast by an impatient
programmer. So for a while, I was using linear cooling. After a while, I realized that the linear cooling
was wasting much time doing pointlessly excessive swapping at high temperatures, and so I came up with
�hyperbolic cooling�, based on an intuition that it had logarithmic cost with respect to initial temperature.
Here is the code:

# A temperature schedule
abstract TempSchedule

# Hyperbolic temperature schedule
immutable HyperTemp <: TempSchedule

cutoff::Int
val::Int

end

# Return number of iterations to run at temperature t for given schedule h
duration( h::HyperTemp, t::Integer ) = t<=h.cutoff ? h.val : round(Int,h.cutoff*h.val/t)

# Do hop! over downward range of temperatures.
function cool!( r::Delacorte, o::Ordering, startTemp::Integer, s::TempSchedule )

for t in startTemp:-1:0
hop!(r,o,t,duration(s,t))

end
end

Julia Mechanics: The notation startTemp:-1:0 denotes a range that counts down from startTemp to
0 with step of -1.

To my chagrin, after the contest �nished, I noticed that my �hyperbolic cooling� is plain old exponential
cooling expressed sideways and di�erentiated. It expresses the number of iterations to run at a given tem-
perature t as c/t, where c = h. cuto� × h.val. The cumulative iterations from temperature a to b has the

form
∫ b

a
c
tdt, which is c log b

a . The cumulative iterations is e�ectively a measure of time. Rewrite to express
temperature in terms of time, and you have classic exponential cooling.

4.6.1 Pro�ling

I used Julia's built-in pro�ler to see where the time was going. Here is an annotated session.

julia> include("delacorte.jl")
N=27
weight built
indexPairArray built

julia> r=randQuad(); # Create random Quad

julia> @profile cool!(randQuad(),up,20000,HyperTemp(100,10000)) # Force compilation

julia> Profile.clear() # Clear profile info

julia> @profile cool!(randQuad(),up,50000,HyperTemp(100,40000)); # Profiling run

julia> Profile.print() # Dump the profile
351 task.jl; anonymous; line: 96
351 REPL.jl; eval_user_input; line: 54
351 loading.jl; include_from_node1; line: 128
351 boot.jl; include; line: 242
350 profile.jl; anonymous; line: 14
350 C:\Delacorte\delacorte.jl; cool!; line: 522 # hop!(r,o,t,duration(s,t))
129 C:\Delacorte\delacorte.jl; hop!; line: 499 # p = getRandomPair()
43 C:\Delacorte\delacorte.jl; rand; line: 79
29 random.jl; rand; line: 229
14 random.jl; gen_rand; line: 81
14 dSFMT.jl; dsfmt_fill_array_close1_open2!; line: 52

19 C:\Delacorte\delacorte.jl; rand; line: 83
99 C:\Delacorte\delacorte.jl; hop!; line: 500 # d = delta(r,p.i,p.j)
8 C:\Delacorte\delacorte.jl; delta; line: 259
1 C:\Delacorte\delacorte.jl; delta; line: 260
6 C:\Delacorte\delacorte.jl; delta; line: 264
28 C:\Delacorte\delacorte.jl; delta; line: 267 # p = (weightSum[i]-weightSum[j])*(∆x*(xj+xi)+∆y*(yj+yi))

- 2((c[i,1]-c[j,1])*∆x + (c[i,2]-c[j,2])*∆y)
46 C:\Delacorte\delacorte.jl; delta; line: 269 # p += 2*weight[i,j]*(∆xˆ2 + ∆yˆ2)
7 C:\Delacorte\delacorte.jl; delta; line: 272
3 C:\Delacorte\delacorte.jl; hop!; line: 501
118 C:\Delacorte\delacorte.jl; hop!; line: 502 # swap!(r,p.i,p.j,d)
2 C:\Delacorte\delacorte.jl; swap!; line: 64
2 C:\Delacorte\delacorte.jl; swap!; line: 277
1 C:\Delacorte\delacorte.jl; swap!; line: 285
100 C:\Delacorte\delacorte.jl; swap!; line: 287 # Bad line info. It’s surely the

@simd loop.
1 C:\Delacorte\delacorte.jl; swap!; line: 290
4 C:\Delacorte\delacorte.jl; swap!; line: 291
1 float.jl; trunc; line: 348

Julia Mechanics: The pro�ler uses statistical sampling. It periodically interrupts the program and records
where the program is currently executing. Each line of pro�le output indicates how much a program spent
where. The �rst number on each line is the how much, in units of samples. The rest of each line describes
where in terms of a �lename, function name, and source line.

The how much is a total of samples for the line and samples for callees, which are shown indented below the
calling line. For example, line 500 in hop! has a total sample count of 99, which includes 8+1+6+28+46+7 =
96 samples for the callee delta and implicitly 3 samples for the line itself.

The pro�le proves the claim in Section 3 that the O(N2) running time of swap! is not a serious bottleneck.12

In fact delta and getRandomPair consume comparable time, because they are more frequently executed
and both have randomly indexed loads that likely cause cache misses.

If you build Julia yourself, you can also pro�le Julia using Intel's R©VTuneTMAmpli�er pro�ler, which has
graphical interface and can drill down to assembly code. See my note �Pro�ling Julia code with Intel VTune
Ampli�er� for details.

4.7 Recooling a Population

Experiments by hand indicated that it was advantageous to iterate applications of cool! and take the best
result. Furthermore, starting with di�erent initial temperatures seemed to help, probably because sometimes
too high a temperature would disturb some useful state, but too low a temperature would miss disturbing
suboptimal state. Later on, I found I could do better by evolving a population of solutions, instead of just
one, through iterated cooling.

Function splat! was engineered to exploit these empirical observations. The name comes from a vision of
frogs hopping away from a starting point long the arms of an asterisk. Given an input population of solutions,
hop! applies cool! to each input solution arity times, starting with temperatures spread out over a range
given by start. The resulting output population is culled to the same size as the input population.

immutable SplatItem
r::Delacorte # Delacorte object
temp::Integer # Initial temperature from which it cooled

end

isless( q::SplatItem, r::SplatItem ) = q.r<r.r

function splat!{T<:Integer}(
outItems::Vector{SplatItem}, inItems::Vector{SplatItem}, o::Ordering,
start::UnitRange{T}, s::TempSchedule, arity::Integer,
verbose::Bool=false

)
!verbose || @printf("splat(%s, %s, %s, %s, %d)\n",

string(inItems), o==up?"up":"down", start, s, arity )
@assert arity≥2 || first(start)==last(start)
slope = (last(start)-first(start))/max(arity-1,1)
for trial=0:arity-1

for i in inItems
r = copy(i.r)
t = round(Int,first(start)+slope*trial)
cool!(r,o,t,s)
# Record temperature only if it improved the individual
push!(outItems, SplatItem(r, lt(o,delacorte(i.r),delacorte(r)) ? t : -1))

end
end
sort!( outItems, alg=MergeSort, rev=(o==up) )
# Remove duplicates

12 The 256-bit wide vector units on the author's MonkeyStation Pro can do eight 32-bit operations in one instruction. Alas
the current Julia JIT used only 128-bit instructions, perhaps because its based on the somewhat dated version 3.3 of LLVM. I
have not measured how much using the full width of the vector units would impact performance.

unique!( outItems )
resize!( outItems, min(length(inItems), length(outItems)) )
saveIfBetter( first(outItems).r, o )

end

Julia Mechanics: The code assumes that isless is already imported (see Section 4.4.3). @printf is
similar to printf in C, except the former checks the format string at compilation time. Functions first
and last extract the �rst and last elements of a range.

The routine saveIfBetter records the best output solution if it is an improvement on the best recorded
solution.

4.8 Automatic Thermostat

I applied splat! by hand many times, in an attempt to �nd a good meta-schedule recipe. What seemed
to work well after an initial �big kick� (high temperature) was repeating with about the same window of
temperatures if things were improving, and otherwise lowering the window. After a while, I tired of doing
this recipe manually, and so created a �robot� routine to do it.

function robo( o::Ordering, r=readBest(o);
first::(Integer,Integer,Integer)=roboFirst,
rest::(Integer,Integer,Integer)=roboRest,
popSize::Integer=roboPop, alpha::FloatingPoint=0.7 )

@assert 0 < alpha < 1
inr = fill(SplatItem(r,-1),popSize)
outr = SplatItem[]
splat!( outr, inr, o, first[1]:first[1], HyperTemp(100,first[2]), first[3], true )
u = round(Int,rest[1]/sqrt(alpha))
while u>0

@assert length(outr)==popSize
outr, inr = inr, outr
resize!(outr,popSize)
for j=1:popSize

outr[j] = SplatItem(inr[j].r,-1)
end
l = u≤30 ? 0 : round(Int,u*alpha)
splat!( outr, inr, o, l:u, HyperTemp(100,rest[2]), rest[3], true )
maxt = maximum(map(i->i.temp,outr))
if maxt≥0

println(" high temp=",maxt)
u = int(maxt/sqrt(alpha))

else
u = l

end
flush(STDOUT)

end
global bg
bg = outr
nothing

end

Julia Mechanics: The type (Integer,Integer,Integer) is a tuple of abstract type Integer. Tuples
have covariant subtyping, so the actual arguments can be any tuple where each component is an Integer.
The assignment outr, inr = inr, outr swaps the two variables. The declaration global bg declares
that variable bg has global scope.

Parameter alpha is the window ratio mentioned in Section 3. The default parameter values roboFirst and

roboRest are global variables that were typically initialized by hand before a run. Saving into global variable
bg is there so that I could recover the most recent output after manually interrupting the program.

The temperature window for each round is l...u. When things are improving, robo raises the temperature
window slightly, which allows it to hunt upwards for a good cooking temperature after the initial �big kick�.
Then the windows usually wanders with a general downward trend, coming down a full window-width
whenever no new competitive solutions are created.

4.9 Top Level

The top level routine flog, not shown, iteratively calls robo. Typically, the steps to run the program were:

1. Start a Julia session

2. Enter const N=size for the problem size of interest.

3. Optionally pin points by setting pinned.

4. Enter include("delacorte.jl")13

5. Set global parameters roboFirst, roboRest, and roboPop.

6. Invoke flog with a direction (up or down) and an iteration count.

Normally, there were 8 sessions going, one for each hardware thread on the system. I kept a spreadsheet
of notes for each size problem. Typically what worked well for a problem also worked well for problems of
similar size.

Pinning corners, or even larger portions of corner regions, was often advantageous because otherwise the
annealing could be trapped into a suboptimal local extrema by a poor choice of corners. Of course pinning
could back�re if the choice was suboptimal. One time I forgot to pin corners for problem for which I was
sure I had the best values, and the program found better corners.

4.9.1 Risky Practice

Often I would set pinned after step 4, and then rerun step 4. This was necessary because some const
variables were computed from pinned, and the compiler can assume that const variables are not rebound
to other objects. So I had to make sure that any function that might depend on those variables was rede�ned,
which was easy to do by re-including the source �le. A better coding alternative would be not rebind the
a�ected variables to change them, but instead update the objects in place.

5 Utilities

Along the way I built some utilities for exploration and tweaking scores. They are in �le util.jl.

1. deopt deoptimizes a solution to a requested amount. I �gured why submit really good solutions early
if that just helps competitors tell if they have really good solutions?

2. tmap shows a map of minimum temperatures required to move each element in a Delacorte matrix.

3. swapSpectrum computes a sorted list of the change in Delacorte number caused by each possible swap.

4. match compares multiple solutions and returns a matrix showing which parts match. The comparison
includes �nding the rotation/�ip of each solution that gives best similarity. The measure of similarity
is biased in favor of matching matrix elements with high weightSum, so as not to be fooled by happen-
stance matchup of low-weight elements. In particular, the matching considers all zero-weight elements
to be equivalent since their positions do not a�ect the Delacorte number.

13 The program �le was actually named f.jl, which saved some typing. Though with Julia's auto-complete, the long name
would not take much more e�ort. Why f.jl? Because it was the successor to e.jl, which was the successor to d.jl, where
the d stood for �Delacorte�. Some successful languages have used the same naming scheme.

6 Conclusion

Julia worked out well for this by combining interactivity with good performance. As important as the code
I've shown was for the contest, equally important is all the experimental code that I did not show because
it turned out to be bad ideas. Part of the power of Julia was letting me get code to the trash can faster.

Early in the contest, I had a multi-process version running, using Julia's support for distributed-memory
parallel processing, but dropped developing it since there were enough separate problems to keep a quad-core
machine busy, and heuristics seemed to trump brute force. But maybe next time around I'll give the parallel
features another whirl.

